
ABSTRACT 
 
The need to calculate statistics for various groups or classifications is ever present. Calculating such statistics may involve 
different strategies with some being less efficient than others. A common approach by new SAS programmers who are 
not very familiar with PROC MEANS is to create a SAS data set for each group of interest and to execute PROC MEANS 
for each group. This strategy can be resource-intensive when large data sets are involved. It requires multiple PROC 
MEANS statements due to multiple input data sets and involves multiple output data sets (one per group of interest). In 
lieu of this, an economy of programming code can be achieved using a simple coding strategy in the DATA step to take 
advantage of PROC MEANS capabilities. Variables that indicate group membership (1 for group membership, blank for 
non-group membership) can be created for each group of interest in a master data set. The master data set with these 
blank/1 indicator variables can then be processed with PROC MEANS and its different statements (i.e., CLASS and 
TYPES) to produce one data set with all the statistics generated for each group of interest.  
 
 
A programmer can calculate disaggregate statistics in a number of different ways. A common approach by new SAS 
programmers who are not very familiar with PROC MEANS is to create a SAS data set for each group of interest and to 
execute PROC MEANS for each group. This strategy can be resource-intensive when large data sets are involved. It 
requires multiple PROC MEANS statements due to multiple input data sets and involves multiple output data sets (one 
per group of interest).  
 

data master; 
input gender $ score; 
cards; 
M 30 
F 100 
; 
 
data males; 
   set master (where=(gender='M')); 
 
data females; 
   set master (where=(gender='F')); 
 
proc means data=males; 
   output out=statsformales; 
 
proc means data=females; 
   output out=statsforfemales; 

 
 

 
 

You could save a little code by using the WHERE statement with PROC MEANS. By doing this, you do not create a data 
set for each group of interest. However, that still leaves you with more than one data set of statistics. 

 
proc means data=master; 
   output out=statsformales; 
   where gender='M'; 
 
proc means data=master; 
   output out=statsforfemales; 
   where gender='F'; 

METHOD 
 
The following discussion will show you how to create blank/1 indicator variables for any group of interest in your master 
data set. Using this single master data set as input, you can obtain a single data set with the disaggregate statistics 
according to these groups of interest from a single PROC MEANS statement. 
 
Consider the following data set. 
 

 
 
 
The following needs to be determined for the raw and scale scores: 

 Average 
 Total number of observations (i.e., denominator of the average) with a score 

According to the following groups: 
1. Males (gender = M) 
2. Females (gender = F) 
3. Students receiving free meals (meals = F) 
4. Students receiving reduced-price meals (meals = R) 
5. Students paying full price for meals (meals = P) 

 
These results can be produced by the following code. The TYPES statement specifies which combinations of the class 
variables are to be used for the calculations.  
 

proc means data=example; 
   class gender meals; 
   var rawscore scalescore; 
   types gender meals; 
   output out=test n=nrs nss mean=mrs mss; 

The corresponding output data set from PROC MEANS is shown below. 
 

 
 
The _TYPE_ variable indicates which CLASS variables produced the data. Although it can be clearly seen from the 
output, as shown above, which levels of each CLASS variable were represented for each line in the data set, the CLASS 
variable configuration can be used to determine the _TYPE_ value. The following example involves two CLASS variables, 
but the concept easily extends to however many CLASS variables there are. 
 

1. Because there are two CLASS variables, think of a binary number that has two digits. The decimal 
equivalents of the binary numbers can be computed as shown. 

Binary Number Decimal Equivalent 
00 0(21) + 0 (20) = 0 
01 0(21) + 1 (20) = 1 
10 1(21) + 0 (20) = 2 
11 1(21) + 1 (20) = 3 

 
2. Gender is the first CLASS variable listed. Use a 0/1 variable based on gender as the first digit in the binary 

number. The digit is: 
0 whenever gender was not used towards the calculations 
1 whenever gender was used towards the calculations 
 

3. Meals is the second CLASS variable listed. Use a 0/1 variable based on meals as the second digit in the 
binary number. The digit is: 

0 whenever meals was not used towards the calculations 
1 whenever meals was used towards the calculations 
 

4. Use the digits from steps #2 and #3 to as digits for the binary number. Since there is a one-to-one 
correspondence between binary numbers and their corresponding decimal equivalents, the value of the 
_TYPE_ variable can only be produced by one binary number. The binary number, using the 1s and 0s as 
defined above, will indicate which CLASS variables went towards a specific record in the output data set. 

 
The _FREQ_ variable is automatically generated by SAS and shows the number of observations for each level of the 
CLASS variable, if there is only one CLASS variable involved. When there is more than one CLASS variable, then it will 
show the number for observations for the combination of levels in the CLASS variables. _FREQ_ , nrs, and nss are not 
necessarily equal as shown in the example. _FREQ_ indicates the number of observations per level or combination of 
levels. The nrs and nss values are the numbers of data points that were included in the calculation of mrs and mss 
respectively. When there are missing values for the raw and scale scores, the values of nrs and nss are less than the 
corresponding  _FREQ_ value. Had there been no missing values, _FREQ_, nrs, and nss would have been all equal to 
each other. 
 
Suppose that a cross-tabulation between gender and meals is required and that missing values for gender and meals 
should be included in the cross-tabulation. Missing values can be included by using the MISSING option with the CLASS 
statement. The NOPRINT option suppresses the output normally generated by the procedure.  
 

proc means nway data=example noprint; 
   class gender meals/missing; 
   var rawscore scalescore; 
   output out=test n=nrs nss mean=mrs mss; 

 
The SUM statement adds up the values of the specified variables and provides the total in the PROC PRINT output; 
 

proc print data=test; 
   sum _freq_ nrs nss; 

Let us suppose the same statistics need to be computed according to the following groups: 

1. All students 
2. Males 
3. Females 
4. Students receiving free or reduced-price meals 
5. Students paying full price for meals 
6. Students receiving free meals 

Note that there is an overlap between categories 4 and 6. 
 
The first step is to create counters for each variable. Counter1 defaults to 1 since this applies to all students. The 
remaining counters are each set to 1 if the criteria for group membership are satisfied. The advantage of using counters 
this way is that they can be easily coded for any group of interest no matter how many variables are used to make the 
determination for group membership (e.g., females receiving free meals, etc.). If it is just counts that are needed, the sum 
of the counters will provide the total number of records per group.  
 

data example; 
set example; 
counter1=1; 
select(gender); 
when ('M') counter2=1; 
when ('F') counter3=1; 
otherwise; 
end; 
**F = free, R = reduced, P = full pay; 
select(meals); 
when ('F') do; counter4=1; counter6=1; end; 
when ('R') counter4=1; 
when ('P') counter5=1; 
otherwise; 
end; 

 
proc means data=example sum maxdec=0; 
   var counter1-counter6; 

There is often a need to compute various types of statistics and not just determine the number of members in groups of 
interest. PROC MEANS can be useful for this purpose. List all the counters in the CLASS and TYPES statements.  
 

proc means data=example; 
   class counter1-counter6/missing; 
   var rawscore scalescore; 
   types counter1 counter2 counter3 counter4 counter5 counter6; 
   output out=test n=nrs nss mean=mrs mss; 

 

 
 
In the output data set, the only rows of interest are the ones with a counter value of 1. The following code will eliminate the 
rows that are not of interest by specifying a WHERE option in the OUTPUT statement. Note that the MISSING option was 
used with the CLASS statement. 
 

proc means data=example; 
   class counter1-counter6/missing; 
   var rawscore scalescore; 
   types counter1 counter2 counter3 counter4 counter5 counter6; 
   output out=test  
      (where=(sum(counter1,counter2,counter3,counter4,counter5,counter6)>0))  
       n=nrs nss mean=mrs mss; 

 

 
 
Because of how PROC MEANS processes data, no observations will result in the data set without the MISSING option as 
shown in the SAS log below. 

Instead of using the TYPES statement shown above, using the “ways 1;” statement instead has the same effect. It tells 
SAS that only levels of single (1) class variables should be considered for the calculations. If the  “ways 2;” statement 
was used, that means the 2-way combinations of all pairs of class variables should be used in the calculations. 
 

proc means data=example; 
   class counter1-counter6/missing; 
   var rawscore scalescore; 
   ways 1; 
   output out=test  
      (where=(sum(counter1,counter2,counter3,counter4,counter5,counter6)>0))  
       n=nrs nss mean=mrs mss; 

 

 
 
This table shows the correspondence between the groups and the _TYPE_ values. Although the example is for six 
counters, the concept can easily be generalized to however many counters are involved. The nth counter will have a 
_TYPE_ value of 2(n–1). 
 

Variable Group Description _TYPE_ Value Binary Number 
counter1 All students 32 = 25 100000 
counter2 Males 16 = 24 010000 
counter3 Females 8 = 23 001000 
counter4 Students receiving free or reduced-price 

meals 
4 = 22 000100 

counter5 Students paying full price for meals 2 = 21 000010 
counter6 Students receiving free meals 1 = 20 000001 

 
 
In looking at the code, the SUM function appears to have many arguments and using 
sum(of counter1-counter6)>0 
instead of  
sum(counter1,counter2,counter3,counter4,counter5,counter6)>0 
is an idea. However, the log shows that there is a problem with this. 

This is a known issue as shown in the SAS Usage Note # 14554. 
 

 
 
 
A macro can be useful when there is a large number of counters. 
 

%macro example(n); 
proc means data=example; 
   class counter1-counter&n/missing; 
   var rawscore scalescore; 
   ways 1; 
   output out=test (where=(sum( 
      %do i=1 %to %eval(&n); 
      counter&i 
      %if &i < %eval(&n) %then %do;,%end;  
      %end; 
      )>0)) n=nrs nss mean=mrs mss; 
%mend example; 
 
%example(6) 

 
When the program is run with the MPRINT system option in effect, the SAS code generated by the macro appears in the 
SAS log. 

